Techniques of Water-Resources Investigations of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

Narrative for Module WEL1RP

This module reads data to build the WELL list.

1. Read ITMP.

(a) If ITMP is less than zero, the well data read for the last stress period will be reused. Print a message to that effect and RETURN.

(b) If ITMP is greater than or equal to zero, it is equal to the number of wells (NWELLS) in the current stress period.

2. If the number of wells (NWELLS) in the current stress period is greater than the number specified as the maximum for the simulation (MXWELL), STOP.

3. Print the number of wells in the current stress period (NWELLS).

4. If there are no wells in the current stress period (NWELLS), bypass further well processing.

5. For each well, read and print the layer, row, column, and well recharge rate.

6. RETURN.

Flow Chart for Module WEL1RP

ITMP is a flag and/or the number of wells. If it is less than zero, it is a flag which indicates that the well data from the last stress period will be reused. If it is greater than or equal to zero, it is the number of wells active during the current stress period.

NWELLS is the number of wells active during the current stress period.

MXWELL is the maximum number of wells which will be active at any one time during the simulation.



```
      SUBROUTINE WEL1RP(WELL,NWELLS,MXWELL,IN,IOUT)
C
C
C-----VERSION 1544 22DEC1982 WEL1RP
C     ******************************************************************
C     READ NEW WELL LOCATIONS AND STRESS RATES
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION WELL(4,MXWELL)
C     ------------------------------------------------------------------
C
C1------READ ITMP(NUMBER OF WELLS OR FLAG SAYING REUSE WELL DATA)
      READ (IN,1) ITMP
    1 FORMAT(I10)
      IF(ITMP.GE.0) GO TO 50
C
C1A-----IF ITMP LESS THAN ZERO REUSE DATA. PRINT MESSAGE AND RETURN.
      WRITE(IOUT,6)
    6 FORMAT(1H0,'REUSING WELLS FROM LAST STRESS PERIOD')
      RETURN
C
C1B-----ITMP=>0.  SET NWELLS EQUAL TO ITMP.
   50 NWELLS=ITMP
      IF(NWELLS.LE.MXWELL) GO TO 100
C
C2------NWELLS>MXWELL.  PRINT MESSAGE. STOP.
      WRITE(IOUT,99) NWELLS,MXWELL
   99 FORMAT(1H0,'NWELLS(',I4,') IS GREATER THAN MXWELL(',I4,')')
      STOP
C
C3------PRINT NUMBER OF WELLS IN CURRENT STRESS PERIOD.
  100 WRITE (IOUT,2) NWELLS
    2 FORMAT(1H0,10X,I4,' WELLS')
C
C4------IF THERE ARE NO ACTIVE WELLS IN THIS STRESS PERIOD THEN RETURN
      IF(NWELLS.EQ.0) GO TO 260
C
C5------READ AND PRINT LAYER,ROW,COLUMN AND RECHARGE RATE.
      WRITE(IOUT,3)
    3 FORMAT(1H ,47X,'LAYER    ROW    COL    STRESS RATE   WELL NO.'/
     1,48X,45('-'))
      DO 250 II=1,NWELLS
      READ (IN,4) K,I,J,Q
    4 FORMAT(3I10,F10.0)
      WRITE (IOUT,5) K,I,J,Q,II
    5 FORMAT(48X,I3,I8,I7,G16.5,I8)
      WELL(1,II)=K
      WELL(2,II)=I
      WELL(3,II)=J
      WELL(4,II)=Q
  250 CONTINUE
C
C6------RETURN
  260 RETURN
      END
```

List of Variables for Module WEL1RP

Variable	Range	Definition
I	Module	Row number of cell containing well.
II	Module	Index for wells.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ITMP	Module	Flag or number of wells. $\geq$ 0, number of wells active during the current stress period. < 0, same wells active during the last stress period will be active during the current stress period.
J	Module	Column number of cell containing well.
K	Module	Layer number of cell containing well.
MXWELL	Package	Maximum number of wells active at any one time.
NWELLS	Package	Number of wells active during the current stress period.
Q	Module	Rate at which the well adds water to the aquifer (negative for discharging well).
WELL	Package	DIMENSION (4,MXWELL), For each well: layer, row, column, and recharge rate of the well.

Narrative for Module WEL1FM

This module adds terms representing well recharge to the accumulator in which the term RHS is formulated.

1. If NWELLS is less than or equal to zero in the current stress period, there are no wells. RETURN.

2. For each well in the WELL list:

(a) If the cell containing the well is external (IBOUND (IC,IR,IL) $\leq$ 0), bypass processing on this well and go on to the next well.

(b) If the cell containing the well is active, subtract the value of Q from the accumulator RHS for that cell.

3. RETURN.

```
      SUBROUTINE WEL1FM(NWELLS,MXWELL,RHS,WELL,IBOUND,
     1        NCOL,NROW,NLAY)
C
C-----VERSION 1233 12MAY1987 WEL1FM
C
C     ******************************************************************
C     SUBTRACT Q FROM RHS
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION RHS(NCOL,NROW,NLAY),WELL(4,MXWELL),
     1            IBOUND(NCOL,NROW,NLAY)
C     ------------------------------------------------------------------
C1------IF NUMBER OF WELLS <= 0 THEN RETURN.
      IF(NWELLS.LE.0) RETURN
C
C2------PROCESS EACH WELL IN THE WELL LIST.
      DO 100 L=1,NWELLS
      IR=WELL(2,L)
      IC=WELL(3,L)
      IL=WELL(1,L)
      Q=WELL(4,L)
C
C2A-----IF THE CELL IS INACTIVE THEN BYPASS PROCESSING.
      IF(IBOUND(IC,IR,IL).LE.0) GO TO 100
C
C2B-----IF THE CELL IS VARIABLE HEAD THEN SUBTRACT Q FROM
C       THE RHS ACCUMULATOR.
      RHS(IC,IR,IL)=RHS(IC,IR,IL)-Q
  100 CONTINUE
C
C3------RETURN
      RETURN
      END
```

List of Variables for Module WEL1FM

Variable	Range	Definition
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
L	Module	Index for wells.
MXWELL	Package	Maximum number of wells active at any one time.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
NWELLS	Package	Number of wells active during the current stress period.
Q	Module	Rate at which the well adds water to the aquifer (negative for discharging wells).
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages.
WELL	Package	DIMENSION (4,MXWELL), For each well: layer, row, column, and recharge rate of the well.

Narrative for Module WEL1BD

This module calculates rates and volumes transferred between the aquifer and wells.

1. Clear the rate accumulators RATIN and RATOUT and the flag (IBD) which indicates that cell-by-cell flow terms should be recorded on a disk.

2. If there are no wells, skip down to step 7.

3. Determine if the cell-by-cell flow terms for wells will be written on a disk. They will be if (1) this is the proper time step (ICBCFL is not equal to zero), (2) if the channel for well-budget terms (IWELCB) is greater than zero, and (3) if the number of wells (NWELLS) is greater than zero.

4. If budget terms are to be written on a disk, set IBD = 1 and clear the buffer (BUFF) in which they will be accumulated.

5. If the number of wells in the current stress period (NWELLS) is not equal to zero, then for each cell in the well list:

   (a) If the cell containing the well is external (IBOUND(I,J,K) $\leq$ 0), bypass further processing of the cell.

   (b) If the user has requested that cell-by-cell rates be printed (IWELCB < 0 and ICBCFL $\neq$ 0), print the rate (Q).

   (c) If the budget terms are to be saved on a disk, add the flow rate (Q) to the buffer (BUFF).

   (d) If Q is positive, add it to RATIN.

   (e) If Q is negative, add it to RATOUT.

6. If the cell-by-cell flow terms are to be recorded, call module UBUDSV to write the contents of buffer (BUFF) onto the disk.

7. Move RATIN and RATOUT into the VBVL array for printing by BAS1OT.

8. Add RATIN and RATOUT multiplied by the time-step length to the volume accumulators in the VBVL array for printing by BAS1OT.

9. Move the well budget term labels to VBNM for printing by BAS1OT.

10. Increment the budget-term counter (MSUM).

11. RETURN.

Flow Chart for Module WEL1BD

RATIN is an accumulator to which all flows into the aquifer are added.

RATOUT is an accumulator to which all flows out of the aquifer are added.

IBD is a flag which, if set, causes cell-by-cell flow terms for well flow to be recorded.

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

EXTERNAL: a cell is said to be external if it is either no flow or constant head (i.e., an equation is not formulated for the cell).

Q is the rate at which the well recharges the aquifer. A discharging well is represented by a negative rate.

IWELCB is a flag and a unit number.

If IWELCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IWELCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IWELCB < 0, well recharge rate will be printed whenever ICBCFL is set.

ICBCFL is a flag.

If ICBCFL $\neq 0$, cell-by-cell flow terms will be either printed or recorded (depending on IWELCB) for the current time step.



```
      SUBROUTINE WEL1BD(NWELLS,MXWELL,VBNM,VBVL,MSUM,WELL,IBOUND,DELT,
     1             NCOL,NROW,NLAY,KSTP,KPER,IWELCB,ICBCFL,BUFF,IOUT)
C
C-----VERSION 1509 12MAY1987 WEL1BD
C     ******************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR WELLS
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 VBNM,TEXT
      DIMENSION VBNM(4,MSUM),VBVL(4,MSUM),WELL(4,MXWELL),
     1          IBOUND(NCOL,NROW,NLAY),BUFF(NCOL,NROW,NLAY)
      DIMENSION TEXT(4)
C
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /'    ','    ','   W','ELLS'/
C     ------------------------------------------------------------------
C
C1------CLEAR RATIN AND RATOUT ACCUMULATORS.
      RATIN=0.
      RATOUT=0.
      IBD=0
C
C2------IF THERE ARE NO WELLS DO NOT ACCUMULATE FLOW
      IF(NWELLS.EQ.0) GO TO 200
C
C3------TEST TO SEE IF CELL-BY-CELL FLOW TERMS WILL BE RECORDED.
      IF(ICBCFL.EQ.0 .OR. IWELCB.LE.0) GO TO 60
C
C4------IF CELL-BY-CELL FLOWS WILL BE SAVED THEN CLEAR THE BUFFER.
      IBD=1
      DO 50 IL=1,NLAY
      DO 50 IR=1,NROW
      DO 50 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
   50 CONTINUE
C
C5------PROCESS WELLS ONE AT A TIME.
   60 DO 100 L=1,NWELLS
      IR=WELL(2,L)
      IC=WELL(3,L)
      IL=WELL(1,L)
      Q=WELL(4,L)
C
C5A-----IF THE CELL IS EXTERNAL IGNORE IT.
```

```
      IF(IBOUND(IC,IR,IL).LE.0)GO TO 100
C
C5B-----PRINT THE INDIVIDUAL RATES IF REQUESTED(IWELCB<0).
      IF(IWELCB.LT.0.AND.ICBCFL.NE.0) WRITE(IOUT,900) (TEXT(N),N=1,4),
     1    KPER,KSTP,L,IL,IR,IC,Q
  900 FORMAT(1H0,4A4,'   PERIOD',I3,'   STEP',I3,'   WELL',I4,
     1     '   LAYER',I3,'   ROW ',I4,'   COL',I4,'   RATE',G15.7)
C
C5C-----IF CELL-BY-CELL FLOWS ARE TO BE SAVED THEN ADD THEM TO BUFFER.
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=BUFF(IC,IR,IL)+Q
      IF(Q) 90,100,80
C
C5D-----PUMPING RATE IS POSITIVE(RECHARGE). ADD IT TO RATIN.
   80 RATIN=RATIN+Q
      GO TO 100
C
C5E-----PUMPING RATE IS NEGATIVE(DISCHARGE). ADD IT TO RATOUT.
   90 RATOUT=RATOUT-Q
  100 CONTINUE
C
C6------IF CELL-BY-CELL FLOWS WILL BE SAVED CALL UBUDSV TO RECORD THEM
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IWELCB,BUFF,NCOL,NROW,
     1                          NLAY,IOUT)
C
C7------MOVE RATES INTO VBVL FOR PRINTING BY MODULE BAS1OT.
  200 VBVL(3,MSUM)=RATIN
      VBVL(4,MSUM)=RATOUT
C
C8------MOVE RATES TIMES TIME STEP LENGTH INTO VBVL ACCUMULATORS.
      VBVL(1,MSUM)=VBVL(1,MSUM)+RATIN*DELT
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
C
C9------MOVE BUDGET TERM LABELS INTO VBNM FOR PRINTING.
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C10-----INCREMENT BUDGET TERM COUNTER(MSUM).
      MSUM=MSUM+1
C
C11-----RETURN
      RETURN
      END
```

List of Variables for Module WEL1BD

Variable	Range	Definition
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
DELT	Global	Length of the current time step.
IBD	Module	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be either printed or recorded (depending on IWELCB) for the current time step.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
IWELCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, well recharge rate will be printed whenever ICBCFL is set.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
L	Module	Index for wells.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
MXWELL	Package	Maximum number of wells active at any one time.

List of Variables for Module WEL1BD (Continued)

Variable	Range	Definition
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
NWELLS	Package	Number of wells active during the current stress period.
Q	Module	Rate at which the well adds water to the aquifer (negative for discharging wells).
RATIN	Module	Accumulator for the total flow into the flow field from wells.
RATOUT	Module	Accumulator for the total flow out of the flow field into wells.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N) Rate for the current time step into the flow field. (2,N) Rate for the current time step out of the flow field. (3,N) Volume into the flow field during simulation. (4,N) Volume out of the flow field during simulation.
WELL	Package	DIMENSION (4,MXWELL), For each well: layer, row, column, and recharge rate of the well.

CHAPTER 9

DRAIN PACKAGE

Conceptualization and Implementation

The Drain Package is designed to simulate the effects of features such as agricultural drains, which remove water from the aquifer at a rate proportional to the difference between the head in the aquifer and some fixed head or elevation, so long as the head in the aquifer is above that elevation, but which have no effect if head falls below that level. The discussion in this section is phrased on the assumption that the features to be simulated are actually agricultural drains.

Figure 39 shows a cross section through a cell, illustrating concepts underlying the simulation of drains in the model. The drain is assumed to run only partially full, so that head within the drain is approximately equal to the median drain elevation, $d_{i,j,k}$. The head computed by the model for cell i,j,k ($h_{i,j,k}$) is actually an average value for the cell, and is normally assumed to prevail at some distance from the drain itself. The drain head, $d_{i,j,k}$ prevails only locally, within the drain--it does not characterize the cell as a whole. Between the drain and the area in which head $h_{i,j,k}$ prevails there exists a radial or semiradial flow pattern in the vertical plane, normally characterized by progressively steeper head gradients as the drain is approached. The head loss within this converging flow pattern forms one part of the head difference $h_{i,j,k}-d_{i,j,k}$. An additional component of head loss may occur in the immediate vicinity of the drain if the hydraulic conductivity in that region differs from the average value used for cell i,j,k--because of the presence of foreign material around



Figure 39.—Cross section through cell i,j,k illustrating head loss in convergent flow into drain.

the drain pipe, or channel-bed material in the case of an open drain (figure 40). Finally, head losses occur through the wall of a drain pipe, depending upon the number and size of the openings in the pipe, and the degree to which those openings may be blocked by chemical precipitates, plant roots, etc.

The three processes discussed above--convergent flow toward the drain, flow through material of different conductivity immediately around the drain, and flow through the wall of the drain--each generate head losses which may be assumed proportional to the discharge, QD, through the system--that is, the discharge from cell i,j,k into the drain. Because these head losses occur in series, the total head loss $h_{i,j,k}-d_{i,j,k}$ may also be taken as proportional to QD. This has been done in the method of simulation embodied in the Drain Package. That is, it has been assumed that the drain function is described by the equation pair

$$QD_{i,j,k} = CD_{i,j,k}(h_{i,j,k} - d_{i,j,k}) \qquad \text{for } h_{i,j,k} > d_{i,j,k} \qquad (69)$$

$$QD_{i,j,k} = 0 \qquad \text{for } h_{i,j,k} \leq d_{i,j,k} \qquad (70)$$

The coefficient $CD_{i,j,k}$ of equation (69) is a lumped (or equivalent) conductance describing all of the head loss between the drain and the region of cell i,j,k in which the head $h_{i,j,k}$ can be the assumed to prevail. It depends on the characteristics of the convergent flow pattern toward the drain, as well as on the characteristics of the drain itself and its immediate environment.

One could attempt to calculate values for CD by developing approximate equations for conductance for the three flow processes, and then calculate the equivalent series conductance. The conductance for each process would be



Figure 40.—Factors affecting head loss immediately around a drain: (a) buried drain pipe in backfilled ditch and (b) open drain.

based on the formulation of a one-dimensional flow equation. The formulations vary significantly depending on the specific drain system being simulated, so no general formulation for calculating CD is presented here. Also, in most situations a specific formulation would require detailed information that is not usually available, such as detailed head distribution around the drain, aquifer hydraulic conductivity near the drain, distribution of the fill material, hydraulic conductivity of fill material, number and size of the drain pipe openings, the amount of clogging materials, and the hydraulic conductivity of the clogging materials. In practice, it is more common to calculate CD from measured values of QD and h-d using equation (69). If h-d is not accurately known, CD is usually adjusted during model calibration in order to match measured values of QD to model calculated values.

Figure 41 shows a graph of QD vs. $h_{i,j,k}$ as defined by equations (69) and (70); the function is similar to that for flow between a surface stream and the aquifer (figure 36) except that flow into the aquifer is excluded, and positive values of QD have been taken as corresponding to flow into the drain. With proper selection of coefficients, the River Package could in fact be utilized to perform the functions of the Drain Package.

Because $QD_{i,j,k}$ in equation (69) has been taken as a flow out of cell i,j,k, it must be subtracted from the left side of equation (24) for each cell affected by a drain, provided the head $h_{i,j,k}$ is above the drain elevation. This is accomplished in the Drain Package by testing to determine whether head exceeds drain elevation, and if so, by adding the term $-CD_{i,j,k}$ to $HCOF_{i,j,k}$ (equation (26)) and adding the term $-CD_{i,j,k}d_{i,j,k}$ to $RHS_{i,j,k}$, as the matrix equations are assembled.



Figure 41.—Plot of flow, QD, into a drain as a function of head, h, in a cell where the elevation of the drain is d and the conductance is CD.

Drain Package Input

Input to the Drain (DRN) Package is read from the unit specified in IUNIT(3).

FOR EACH SIMULATION

DRN1AL

1. Data: MXDRN IDRNCB
   Format: I10 I10

FOR EACH STRESS PERIOD

DRN1RP

2. Data: ITMP
   Format: I10

3.

	Layer	Row	Col	Elevation	Cond
Data:	Layer	Row	Col	Elevation	Cond
Format:	I10	I10	I10	F10.0	F10.0

(Input item 3 normally consists of one record for each drain. If ITMP is negative or zero, item 3 will not be read.)

Explanation of Fields Used in Input Instructions

MXDRN--is the maximum number of drain cells active at one time.

IDRNCB--is a flag and a unit number.

If IDRNCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If IDRNCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IDRNCB < 0, drain leakage for each cell will be printed whenever ICBCFL is set.

ITMP--is a flag and a counter.

If ITMP < 0, drain data from the last stress period will be reused.

If ITMP $\geq$ 0, ITMP will be the number of drains active during the current stress period.

Layer--is the layer number of the cell containing the drain.

Row--is the row number of the cell containing the drain.

Column--is the column number of the cell containing the drain.

Elevation--is elevation of the drain.

Cond--is the hydraulic conductance of the interface between the aquifer and the drain.

SAMPLE INPUT TO THE DRAIN PACKAGE

DATA ITEM	EXPLANATION	INPUT RECORDS				
1	{MXDRNR, IDRNCB}	3	55			
2	{ITMP} FOR FIRST STRESS PERIOD	3				
3	{LAYER, ROW, COLUMN, ELEVATION, COND } FOR FIRST DRAIN	2	6	4	220.	.7
3	{LAYER, ROW, COLUMN, ELEVATION, COND } FOR SECOND DRAIN	2	7	4	225.	.9
3	{LAYER, ROW, COLUMN, ELEVATION, COND } FOR THIRD DRAIN	2	5	4	210.	.8
2	{ITMP} FOR SECOND STRESS PERIOD	-1				
2	{ITMP} FOR THIRD STRESS PERIOD	-1				
2	{ITMP} FOR FOURTH STRESS PERIOD	2				
3	{LAYER, ROW, COLUMN, ELEVATION, COND } FOR FIRST DRAIN	2	5	4	210.	.8
3	{LAYER, ROW, COLUMN, ELEVATION, COND } FOR SECOND DRAIN	2	6	4	220.	.7
2	{ITMP} FOR FIFTH STRESS PERIOD	0				
2	{ITMP} FOR SIXTH STRESS PERIOD	-1				

Module Documentation for the Drain Package

The Drain Package (DRN1) consists of four modules, all of which are called by the MAIN program. The modules are:

DRN1AL Allocates space for an array that contains the drain list (DRAI).

DRN1RP Reads location, drain elevation, and drain conductance of each cell containing a drain.

DRN1FM Adds the terms $-CD_{i,j,k}$ and $-CD_{i,j,k}d_{i,j,k}$ to the accumulators $HCOF_{i,j,k}$ and $RHS_{i,j,k}$, respectively.

DRN1BD Calculates the rates and accumulated volume of drainage from the flow system.

Narrative for Module DRN1AL

This module allocates space in the X array to store the list of drains.

1. Print a message identifying the package and initialize NDRAIN (number of drains).

2. Read and print MXDRAN (the maximum number of drains) and IDRNCB (the file number for saving cell-by-cell flow terms or a flag indicating that cell-by-cell flow terms should be printed).

3. Set LCDRAI (which will point to the first element in the drain list) equal to ISUM (which points to the first unallocated element in the X array.

4. Calculate the amount of space needed for the drain list (five values for each drain--row, column, layer, drain elevation, and drain conductance).

5. Print the number of elements in the X array used by the Drain Package.

6. RETURN.

Flow Chart for Module DRN1AL

NDRAIN is the number of drains being simulated at any given time.

MXDRN is the maximum number of drains simulated.

IDRNCB is a flag and a unit number.

If IDRNCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IDRNCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IDRNCB < 0, drain leakage for each drain will be printed whenever ICBCFL is set.

LCDRAI is the location, in the X array, of the list of drain data (DRAI).



```
      SUBROUTINE DRN1AL(ISUM,LENX,LCDRAI,NDRAIN,MXDRN,IN,IOUT,
     1                          IDRNCB)
C
C-----VERSION 1604 12MAY1987 DRN1AL
C     ******************************************************************
C     ALLOCATE ARRAY STORAGE FOR DRAIN PACKAGE
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
C     ------------------------------------------------------------------
C
C1------IDENTIFY PACKAGE AND INITIALIZE NDRAIN.
      WRITE(IOUT,1)IN
    1 FORMAT(1H0,'DRN1 -- DRAIN PACKAGE, VERSION 1, 9/1/87',
     1' INPUT READ FROM UNIT',I3)
      NDRAIN=0
C
C2------READ & PRINT MXDRN & IDRNCB(UNIT & FLAG FOR CELL-BY-CELL FLOW)
      READ(IN,2) MXDRN,IDRNCB
    2 FORMAT(2I10)
      WRITE(IOUT,3) MXDRN
    3 FORMAT(1H ,'MAXIMUM OF',I5,' DRAINS')
      IF(IDRNCB.GT.0) WRITE(IOUT,9) IDRNCB
    9 FORMAT(1X,'CELL-BY-CELL FLOWS WILL BE RECORDED ON UNIT',I3)
      IF(IDRNCB.LT.0) WRITE(IOUT,8)
    8 FORMAT(1X,'CELL-BY-CELL FLOWS WILL BE PRINTED WHEN ICBCFL NOT 0')
C
C3------SET LCDRAI EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
      LCDRAI=ISUM
C
C4------CALCULATE AMOUNT OF SPACE USED BY THE DRAIN PACKAGE.
      ISP=5*MXDRN
      ISUM=ISUM+ISP
C
C5------PRINT AMOUNT OF SPACE USED BY DRAIN PACKAGE.
      WRITE(IOUT,4) ISP
    4 FORMAT(1X,I8,' ELEMENTS IN X ARRAY ARE USED FOR DRAINS')
      ISUM1=ISUM-1
      WRITE(IOUT,5) ISUM1,LENX
    5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
      IF(ISUM1.GT.LENX) WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE DIMENSIONED LARGER***')
C
C6------RETURN
      RETURN
      END
```

List of Variables for Module DRN1AL

Variable	Range	Definition
IDRNCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will be neither printed nor recorded. < 0, leakage for each drain will be printed.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISP	Module	Number of words in the X array allocated by this module.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	ISUM - 1.
LCDRAI	Package	Location in the X array of the first element of array DRAI.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN Program.
MXDRN	Package	Maximum number of drains active at any one time.
NDRAIN	Package	Number of drains active during the current stress period.